Lightweight Implementation of Generics and Dynamics

Johan Jeuring

Utrecht University

17 February, 2014

Libraries for Generic Programming in Haskell

- Haskell is powerful enough to support most generic programming concepts by means of a library.
- Compared with a language extension (PolyP, Generic Haskell), a library is much easier to ship, support, and maintain.
- Compared with a preprocessing tool like DrIFT or Template Haskell, a library gives you much more support, such as types. Of course, a library might be accompanied by tools.

GP Libraries in Haskell (1)

- Lightweight Implementation of Generics and Dynamics (2002)
- Strafunski (2002)
- Scrap Your Boilerplate (SYB) (2003,2004,2005)
- Prototyping Generic Programming using Template Haskell (2004)
- Generics for the Masses (2004)
- SYB Reloaded, Revolutions (2006)
- Generic programming, now! (2006)
- RepLib (2006)
- Smash Your Boilerplate (2006)
- Almost Compositional Functions (2006)
- Extensible and Modular Generics for the Masses (2006)
- Uniplate (2007)

GP Libraries in Haskell (2)

- Alloy (2008)
- Instant Generics (2009)
- Multirec (2009)
- Regular (2009)
- Pointless Haskell (2010)
- Generic Deriving (2010)
- Multiplate (2010)
- Yoko (2012)
- Shapely-data (2013)
- ...

Essential Concepts In Libraries

There are three essential concepts related to generic programming which we will discuss for each library:

- Run-time type representation
- Generic view on data
- Support for overloading

Equality

Equality is a classic generic programming example.

$$
\begin{aligned}
&\mathrm{eq}_{\mathrm{String}} :: \mathrm{String} \to \mathrm{String} \to \mathrm{Bool} \\
&\mathrm{eq}_{\mathrm{String}}\ [\,]\ \ [\,]\ \ = \mathrm{True} \\
&\mathrm{eq}_{\mathrm{String}}\ [\,]\ \ \_\ \ = \mathrm{False} \\
&\mathrm{eq}_{\mathrm{String}}\ \_\ \ [\,]\ \ = \mathrm{False} \\
&\mathrm{eq}_{\mathrm{String}}\ (a : as)\ (b : bs) = a \equiv b \wedge \mathrm{eq}_{\mathrm{String}}\ as\ bs
\end{aligned}
$$

The algorithm is simple:

- Check whether two values are in the same alternative.
- If not, they are not equal.
- Otherwise, they are equal if all arguments are equal.

LIGD

A Lightweight Implementation of Generics and Dynamics ("LIGD"):

- An approach to embedding generic functions and dynamic values into Haskell 98 augmented with existential types.
- Reflect the type argument onto the value level so we can do ordinary pattern matching on types.
- Developed by James Cheney and Ralf Hinze.
- We describe a variant of LIGD that uses GADTs.
- We do not discuss the "dynamics" feature of the library.

Example: Equality

```
geq :: Rep a →          a →          a →          Bool
geq    RUnit            Unit         Unit         = True
geq    RInt             i            j            = i ≡ j
geq    RChar            c            d            = c ≡ d
geq    (RSum r_a _)     (L a_1)      (L a_2)      = geq r_a a_1 a_2
geq    (RSum _ r_b)     (R b_1)      (R b_2)      = geq r_b b_1 b_2
geq    (RSum _ _)       _            _            = False
geq    (RProd r_a r_b)  (a_1 :×: b_1) (a_2 :×: b_2) = geq r_a a_1 a_2 ∧ geq r_b b_
```

This is called a type-indexed function.

Structure types

These types represent some of the basic structural elements of datatypes.

Question

What are equivalent *standard* types?

```
data Unit = Unit
```

```
data () = ()
```

```
data a :+: b = L a | R b
```

```
data Either a b = Left a | Right b
```

```
data a :×: b = a :×: b
```

```
data (a, b) = (a, b)
```

Run-Time Type Representation

LIGD uses a GADT for representing the structure of a type at run-time:

```
data Rep :: ∗ → ∗ where
  RUnit :: Rep Unit
  RInt  :: Rep Int
  RChar :: Rep Char
  RSum  :: Rep a → Rep b → Rep (a :+: b)
  RProd :: Rep a → Rep b → Rep (a :×: b)
```

`Rep t` is the type representation of `t`.

Question

What purpose do the type indexes serve here?

Going Generic: Universe Extension

The structure of a user-defined datatype `t` is represented by the following `Rep` constructor.

```
data Rep :: * → * where
    ...
  RType :: EP t r → Rep r → Rep t
```

The type `r` is the structure representation type of `t`, where `r` is a type isomorphic to `t`.

The isomorphism is witnessed by an embedding projection pair:

```
data EP t r = EP {from :: (t → r), to :: (r → t)}
```

Structure of Lists (1)

The structure representation type of `List` :

```
data List  a = Nil   | Cons a     (List a)
type List_s a = Unit :+:      a :×: List a
```

We define two functions to transform between the Haskell type and its LIGD representation.

```
fromList :: List a → List_s a
```

```
toList :: List_s a → List a
```

```
fromList Nil           = L Unit
fromList (Cons a as) = R (a :×: as)
```

```
toList (L Unit)       = Nil
toList (R (a :×: as)) = Cons a as
```

Structure of Lists (2)

We define an embedding-projection pair to implement the transformation between the Haskell type and its LIGD representation.

```
rList :: Rep a → Rep (List a)
rList r_a = RType (EP fromList toList)
                  (RSum RUnit (RProd r_a (rList r_a)))
```

Generic Equality

`geq` is turned into a generic function by adding the following case to its definition.

```
geq :: Rep a → a → a → Bool
 ...
geq (RType ep r_a) t1 t2 = geq r_a (from ep t1) (from ep t2)
```

Notes on `geq`

- `geq` can be viewed as an implementation of **deriving** Eq in Haskell.
- Similarly, we can define functions that implement the methods of the other classes that can be derived in Haskell: `Show`, `Read`, `Ord`, `Enum`, and `Bounded`.

Generic Show

We will implement a generic `show` function to illustrate how a library deals with:

- Constructor names
- Ad-hoc cases for particular datatypes: `"abc"` should not be printed as `Cons 'a' (Cons 'b' (Cons 'c' Nil))`

Constructor Names

To deal with constructor names, we add a constructor to `Rep`.

```
data Rep :: ∗ → ∗ where
  ...
  RCon :: String → Rep a → Rep a
```

Using this constructor, the representation of `List` becomes:

```
rList :: Rep a → Rep (List a)
rList r_a = RType (EP fromList toList)
                  (RSum (RCon "Nil" RUnit)
                        (RCon "Cons" (RProd r_a (rList r_a))))
```

Generic Show: First Attempt

```
gshow :: Rep t → t → String
gshow RInt              t       = show t
gshow RChar             t       = show t
gshow RUnit             t       = ""
gshow (RSum r_a _)      (L a)   = gshow r_a a
gshow (RSum _ r_b)      (R b)   = gshow r_b b
gshow (RProd r_a r_b)   (a :×: b) = gshow r_a a ++ "  " ++ gshow r_b b
gshow (RType ep r_a)    t       = gshow r_a (from ep t)
gshow (RCon s RUnit)    t       = s
gshow (RCon s r_a)      t       = "(" ++ s ++ "  " ++ gshow r_a t ++ ")"
```

Question

What will `gp` look like?

```
gp = gshow (rList RChar) (Cons 'g' (Cons 'p' Nil))
```

Support for Overloading (1)

- This definition shows strings and Haskell's lists using constructor names.
- We want `gshow` to use the standard Haskell `"string"` format instead.
- We want `gshow` to behave in a *specialized*, *non-generic* way for strings.
- Solution: Extend `Rep` with a case for strings:

```
data Rep :: * → * where
  ...
  RString :: Rep String
```

Support for Overloading (2)

Now we can add the following line to the generic show function to obtain type-specific behavior for the type `String`.

```
gshow :: Rep t → t → String
 ...
gshow RString s = s
```

Support for Overloading Is Weak

- We have to adapt the type representation type `Rep` to obtain type-specific behavior in the `gshow` function.
- It is undesirable to adapt a library for the purpose of obtaining special behavior of a single generic function on a particular datatype.
- Unfortunately, this is unavoidable in the LIGD library.
- This implies that many users will construct their own variant of the LIGD library, making both the library and the generic functions written using it less portable and reusable.

Producer Function: Generic Empty

- Both `geq` and `gshow` are generic consumer functions. They take generic values as arguments and produce non-generic results.
- We can also define generic producer functions that do the reverse.
- Simple example: associate an "empty" value with every type.

```
gempty :: Rep a → a
gempty RUnit           = Unit
gempty RInt            = 0
gempty RChar           = '\NUL'
gempty (RSum r_a _)    = L (gempty r_a)
gempty (RProd r_a r_b) = gempty r_a :×: gempty r_b
gempty (RType ep r_a)  = to ep (gempty r_a)
gempty (RCon  s r_a)   = gempty r_a
```

Generic Flatten (1)

- Many datatypes can be considered container datatypes: those used to store and structure values.
- Examples are the datatypes `List a`, `Maybe a`, etc.
- A function `flatten` takes a value of a container datatype and returns a list containing all values that it contains.
- `Data.Tree` has `flatten :: Tree a → [a]`.
- The `Prelude` has a related function for lists of lists, `concat :: [[a]] → [a]`.

Generic Flatten (2)

Question

What is the type of the generic flatten function?

Attempt 1:

```
gflatten :: Rep g → g a → [a]
```

Attempt 2:

```
gflatten :: Rep (g a) → g a → [a]
```

Attempt 3:

```
gflatten :: Rep1 ... a → b → [a]
```

A New Representation Type

```
data Rep1 :: (* → *) → * → * where
  RChar1 ::                                  Rep1 g Char
  RInt1  ::                                  Rep1 g Int
  RUnit1 ::                                  Rep1 g Unit
  RSum1  :: Rep1 g a → Rep1 g b →            Rep1 g (a :+: b)
  RProd1 :: Rep1 g a → Rep1 g b →            Rep1 g (a :×: b)
  RCon1  :: String → Rep1 g a →              Rep1 g a
  RType1 :: EP b a → Rep1 g a →              Rep1 g b
  RVar1  :: g a →                            Rep1 g a
```

We added a new constructor `RVar1` for the type of the container elements.

A New List Representation

```
rList1 :: Rep1 g a → Rep1 g (List a)
rList1 r_a = RType1 (EP fromList toList)
                    (RSum1 RUnit1 (RProd1 r_a (rList1 r_a)))
```

Defining GFlatten (1)

```
newtype GFlatten b a = GFlatten {selFlatten :: a → [b]}
```

```
gflatten' RUnit1           Unit      = []
gflatten' (RSum1 r_a _)    (L a)     = gflatten' r_a a
gflatten' (RSum1 _ r_b)    (R b)     = gflatten' r_b b
gflatten' (RProd1 r_a r_b) (a :×: b) = gflatten' r_a a ++ gflatten' r_b b
gflatten' RInt1            i         = []
gflatten' RChar1           c         = []
gflatten' (RCon1 _ r_a)    x         = gflatten' r_a x
gflatten' (RType1 ep r_a)  x         = gflatten' r_a (from ep x)
gflatten' (RVar1 f)        x         = selFlatten f x
```

The type:

```
gflatten' :: Rep1 (GFlatten b) a → a → [b]
```

Defining GFlatten (2)

To simplify the types, we define a type synonym:

```
type GFlattenRep a b = Rep1 (GFlatten b) a
```

```
gflatten' :: GFlattenRep a b → a → [b]
```

Defining GFlatten (3)

We need to describe:

1. A GFlattenRep for containers
2. What to do with the elements of the container

Recall the representation for lists:

```
rList1 :: Rep1 g a → Rep1 g (List a)
```

We use a function as a representation for containers...

```
gflatten :: (GFlattenRep a a → GFlattenRep b c) → b → [c]
gflatten repContainer = gflatten′ (repContainer repVar)
```

... and we tell it to insert every element into a singleton list.

```
where repVar :: GFlattenRep a a
      repVar = RVar1 (GFlatten (:[]))
```

Defining GFlatten (4)

To obtain an instance of the generic function `gflatten'` on `List`, we write:

```
flattenList :: List a → [a]
flattenList = gflatten rList1
```

Generic Map (1)

The generic map function `gmap`:

- A generalisation of the `map` on lists
- Takes a function of type `a → b` and a value of a datatype containing `a`-type elements and applies the function to all the elements in the value.
- `gmap` can be viewed as the implementation of **`deriving`** for the `Functor` type class in Haskell.
- As with `gflatten`, `gmap` needs to know where the occurrences of the type argument of the datatype appear in a constructor.

Generic Map (2)

- Suppose we use the representation type `Rep1` to implement the generic map function.
- The action on variables then has type `g a → Rep1 g a`, with `g` instantiated to a `newtype GMap`.
- The argument function of `gmap` can only return a value of a type that depends on `a`, or a constant type, but *not* a value of a type `b`.
- To pass a function of type `a → b`, we need an extra type variable in the `RVar1` constructor.

Another Type Representation

```
data Rep2 :: (* → * → *) → * → * → * where
  RChar2 ::                                          Rep2 g Char Char
  RInt2  ::                                          Rep2 g Int Int
  RUnit2 ::                                          Rep2 g Unit Unit
  RSum2  :: Rep2 g a b → Rep2 g c d →                Rep2 g (a :+: c) (b :+: d)
  RProd2 :: Rep2 g a b → Rep2 g c d →                Rep2 g (a :×: c) (b :×: d)
  RCon2  :: String → Rep2 g a b →                    Rep2 g a b
  RType2 :: EP b a → EP d c → Rep2 g a c →           Rep2 g b d
  RVar2  :: g a b →                                  Rep2 g a b
```

`RVar2` represents the two type variables.

Another List Representation

```
rList2 :: Rep2 g a b → Rep2 g (List a) (List b)
rList2 r_a = RType2 (EP fromList toList)
                    (EP fromList toList)
                    (RSum2 RUnit2 (RProd2 r_a (rList2 r_a)))
```

Defining GMap (1)

```
newtype GMap a b = GMap {selMap :: a → b}
type GMapRep a b = Rep2 GMap a b
```

```
gmap′ :: GMapRep a b → a → b
gmap′ RUnit2                  Unit      = Unit
gmap′ (RSum2 r_a _)           (L a)     = L (gmap′ r_a a)
gmap′ (RSum2 _ r_b)           (R b)     = R (gmap′ r_b b)
gmap′ (RProd2 r_a r_b)        (a :×: b) = gmap′ r_a a :×: gmap′ r_b b
gmap′ RInt2                   i         = i
gmap′ RChar2                  c         = c
gmap′ (RCon2 _ r_a)           x         = gmap′ r_a x
gmap′ (RType2 ep1 ep2 r_a)    x         = (to ep2 ∘ gmap′ r_a ∘ from ep1) x
gmap′ (RVar2 f)               x         = selMap f x
```

Defining GMap (2)

`gmap` is defined similarly to `gflatten`.

```
gmap :: (GMapRep a b → GMapRep c d) → (a → b) → c → d
gmap repContainer f = gmap′ (repContainer repVar)
  where repVar = RVar2 (GMap f)
```

Unlike `gflatten`, we use a parameter for the function `f` on the elements.

Defining GMap (3)

Recall the `List` representation:

```
rList2 :: Rep2 g a b → Rep2 g (List a) (List b)
```

with the instance:

```
rList2 :: GMapRep a b → GMapRep (List a) (List b)
```

To obtain an instance of `gmap` on the datatype `List a` we write:

```
gmapList :: (a → b) → List a → List b
gmapList = gmap rList2
```

Notes on GMap

Question

Do we need to introduce a new representation type for every generic function we define?

- For all practical purposes, it appears that we need at most 3 type variables.
- We could use the datatype `Rep3` for all of our generic functions, but that would introduce many type variables which are never used.

Conclusions

We have covered:

- A library for generic programming in Haskell: LIGD
- The important aspects, strengths, and weaknesses of this library.

LIGD:

- Is just one of many libraries for generic programming.
- Uses one of a number of views on datatype structure.
- May be one of the easiest to implement and understand.
- May be one of the least modular and extensible.